

NAG Library Routine Document

D02NBF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

1 Purpose

D02NBF is a forward communication routine for integrating stiff systems of explicit ordinary differential equations when the Jacobian is a full matrix.

2 Specification

```

SUBROUTINE D02NBF (NEQ, LDYSAV, T, TOUT, Y, YDOT, RWORK, RTOL, ATOL, ITOL,      &
                  INFORM, FCN, YSAV, SDYSAV, JAC, WKJAC, NWKJAC, MONITR,      &
                  ITASK, ITRACE, IFAIL)

INTEGER           NEQ, LDYSAV, ITOL, INFORM(23), SDYSAV, NWKJAC, ITASK,      &
                  ITRACE, IFAIL
REAL (KIND=nag_wp) T, TOUT, Y(NEQ), YDOT(NEQ), RWORK(50+4*NEQ), RTOL(*),    &
                  ATOL(*), YSAV(LDYSAV,SDYSAV), WKJAC(NWKJAC)
EXTERNAL          FCN, JAC, MONITR

```

3 Description

D02NBF is a general purpose routine for integrating the initial value problem for a stiff system of explicit ordinary differential equations,

$$y' = g(t, y).$$

It is designed specifically for the case where the Jacobian $\frac{\partial g}{\partial y}$ is a full matrix.

Both interval and step oriented modes of operation are available and also modes designed to permit intermediate output within an interval oriented mode.

An outline of a typical calling program for D02NBF is given below. It calls the full matrix linear algebra setup routine D02NSF, the Backward Differentiation Formula (BDF) integrator setup routine D02NVF, and its diagnostic counterpart D02NYF.

```

!      Declarations

EXTERNAL FCN, JAC, MONITR
.
.
IFAIL = 0
CALL D02NVF(...,IFAIL)
CALL D02NSF(NEQ, LDYSAV, JCEVAL, NWKJAC, RWORK, IFAIL)
IFAIL = -1
CALL D02NBF(NEQ, LDYSAV, T, TOUT, Y, YDOT, RWORK, RTOL,      &
           ATOL, ITOL, INFORM, FCN, YSAVE, SDYSAV, JAC, WKJAC, NWKJAC, &
           MONITR, ITASK, ITRACE, IFAIL)
IF (IFAIL.EQ.1 .OR. IFAIL.GE.14) STOP
IFAIL = 0
CALL D02NYF(...)
.
.
STOP
END

```

The linear algebra setup routine D02NSF and one of the integrator setup routines, D02NVF or D02NWF, must be called prior to the call of D02NBF. The integrator diagnostic routine D02NYF may be called after the call to D02NBF. There is also a routine, D02NZF, designed to permit you to change step size on a continuation call to D02NBF without restarting the integration process.

4 References

See the D02M–N sub-chapter Introduction.

5 Parameters

- 1: NEQ – INTEGER *Input*
On entry: the number of differential equations to be solved.
Constraint: $NEQ \geq 1$.

- 2: LDYSAV – INTEGER *Input*
On entry: a bound on the maximum number of differential equations to be solved during the integration.
Constraint: $LDYSAV \geq NEQ$.

- 3: T – REAL (KIND=nag_wp) *Input/Output*
On entry: t , the value of the independent variable. The input value of T is used only on the first call as the initial point of the integration.
On exit: the value at which the computed solution y is returned (usually at TOUT).

- 4: TOUT – REAL (KIND=nag_wp) *Input*
On entry: the next value of t at which a computed solution is desired. For the initial t , the input value of TOUT is used to determine the direction of integration. Integration is permitted in either direction (see also ITASK).
Constraint: $TOUT \neq T$.

- 5: Y(NEQ) – REAL (KIND=nag_wp) array *Input/Output*
On entry: the values of the dependent variables (solution). On the first call the first NEQ elements of Y must contain the vector of initial values.
On exit: the computed solution vector, evaluated at T (usually $T = TOUT$).

- 6: YDOT(NEQ) – REAL (KIND=nag_wp) array *Output*
On exit: the time derivatives y' of the vector y at the last integration point.

- 7: RWORK(50 + 4 × NEQ) – REAL (KIND=nag_wp) array *Communication Array*

- 8: RTOL(*) – REAL (KIND=nag_wp) array *Input*
Note: the dimension of the array RTOL must be at least 1 if ITOL = 1 or 2, and at least NEQ otherwise.
On entry: the relative local error tolerance.
Constraint: $RTOL(i) \geq 0.0$ for all relevant i (see ITOL).

9: ATOL(*) – REAL (KIND=nag_wp) array *Input*

Note: the dimension of the array ATOL must be at least 1 if ITOL = 1 or 3, and at least NEQ otherwise.

On entry: the absolute local error tolerance.

Constraint: $ATOL(i) \geq 0.0$ for all relevant i (see ITOL).

10: ITOL – INTEGER *Input*

On entry: a value to indicate the form of the local error test. ITOL indicates to D02NBF whether to interpret either or both of RTOL or ATOL as a vector or a scalar. The error test to be satisfied is $\|e_i/w_i\| < 1.0$, where w_i is defined as follows:

ITOL	RTOL	ATOL	w_i
1	scalar	scalar	$RTOL(1) \times y_i + ATOL(1)$
2	scalar	vector	$RTOL(1) \times y_i + ATOL(i)$
3	vector	scalar	$RTOL(i) \times y_i + ATOL(1)$
4	vector	vector	$RTOL(i) \times y_i + ATOL(i)$

e_i is an estimate of the local error in y_i , computed internally, and the choice of norm to be used is defined by a previous call to an integrator setup routine.

Constraint: ITOL = 1, 2, 3 or 4.

11: INFORM(23) – INTEGER array *Communication Array*

12: FCN – SUBROUTINE, supplied by the user. *External Procedure*

FCN must evaluate the derivative vector for the explicit ordinary differential equation system, defined by $y' = g(t, y)$.

The specification of FCN is:

```
SUBROUTINE FCN (NEQ, T, Y, F, IRES)
  INTEGER          NEQ, IRES
  REAL (KIND=nag_wp) T, Y(NEQ), F(NEQ)
```

1: NEQ – INTEGER *Input*

On entry: the number of differential equations being solved.

2: T – REAL (KIND=nag_wp) *Input*

On entry: t , the current value of the independent variable.

3: Y(NEQ) – REAL (KIND=nag_wp) array *Input*

On entry: the value of y_i , for $i = 1, 2, \dots, NEQ$.

4: F(NEQ) – REAL (KIND=nag_wp) array *Output*

On exit: the value y'_i , given by $y'_i = g_i(t, y)$, for $i = 1, 2, \dots, NEQ$.

5: IRES – INTEGER *Input/Output*

On entry: IRES = 1.

On exit: you may set IRES as follows to indicate certain conditions in FCN to the integrator:

IRES = 1

Indicates a normal return from FCN, that is IRES has not been altered by you and integration continues.

IRES = 2

Indicates to the integrator that control should be passed back immediately to the calling (sub)program with the error indicator set to IFAIL = 11.

IRES = 3

Indicates to the integrator that an error condition has occurred in the solution vector, its time derivative or in the value of t . The integrator will use a smaller time step to try to avoid this condition. If this is not possible the integrator returns to the calling (sub)program with the error indicator set to IFAIL = 7.

IRES = 4

Indicates to the integrator to stop its current operation and to enter MONITR immediately with parameter IMON = -2.

FCN must either be a module subprogram USED by, or declared as EXTERNAL in, the (sub)program from which D02NBF is called. Parameters denoted as *Input* must **not** be changed by this procedure.

- 13: YSAV(LDYSAV,SDYSAV) – REAL (KIND=nag_wp) array *Communication Array*
 14: SDYSAV – INTEGER *Input*

On entry: the second dimension of the array YSAV as declared in the (sub)program from which D02NBF is called. An appropriate value for SDYSAV is described in the specification of the integrator setup routines D02NVF and D02NWF. This value must be the same as that supplied to the integrator setup routine.

- 15: JAC – SUBROUTINE, supplied by the NAG Library or the user. *External Procedure*

JAC must evaluate the Jacobian of the system. If this option is not required, the actual argument for JAC must be the dummy routine D02NBZ. (D02NBZ is included in the NAG Library.) You must indicate to the integrator whether this option is to be used by setting the parameter JCEVAL appropriately in a call to the full linear algebra setup routine D02NSF.

First we must define the system of nonlinear equations which is solved internally by the integrator. The time derivative, y' , generated internally, has the form

$$y' = (y - z)/(hd),$$

where h is the current step size and d is a parameter that depends on the integration method in use. The vector y is the current solution and the vector z depends on information from previous time steps. This means that $\frac{d}{dy}(\cdot) = (hd)\frac{d}{dy}(\cdot)$. The system of nonlinear equations that is solved has the form

$$y' - g(t, y) = 0$$

but it is solved in the form

$$r(t, y) = 0,$$

where r is the function defined by

$$r(t, y) = (hd)((y - z)/(hd) - g(t, y)).$$

It is the Jacobian matrix $\frac{\partial r}{\partial y}$ that you must supply in JAC as follows:

$$\frac{\partial r_i}{\partial y_j} = 1 - (hd)\frac{\partial g_i}{\partial y_j}, \quad \text{if } i = j,$$

$$\frac{\partial r_i}{\partial y_j} = -(hd)\frac{\partial g_i}{\partial y_j}, \quad \text{otherwise.}$$

The specification of JAC is:

```
SUBROUTINE JAC (NEQ, T, Y, H, D, P)
```

```
INTEGER          NEQ
```

```
REAL (KIND=nag_wp) T, Y(NEQ), H, D, P(NEQ,NEQ)
```

1: NEQ – INTEGER *Input*

On entry: the number of differential equations being solved.

2: T – REAL (KIND=nag_wp) *Input*

On entry: t , the current value of the independent variable.

3: Y(NEQ) – REAL (KIND=nag_wp) array *Input*

On entry: y_i , for $i = 1, 2, \dots, \text{NEQ}$, the current solution component.

4: H – REAL (KIND=nag_wp) *Input*

On entry: the current step size.

5: D – REAL (KIND=nag_wp) *Input*

On entry: the parameter d which depends on the integration method.

6: P(NEQ,NEQ) – REAL (KIND=nag_wp) array *Input/Output*

On entry: is set to zero.

On exit: $P(i, j)$ must contain $\frac{\partial r_i}{\partial y_j}$, for $i = 1, 2, \dots, \text{NEQ}$ and $j = 1, 2, \dots, \text{NEQ}$.

Only the nonzero elements of this array need be set, since it is preset to zero before the call to JAC.

JAC must either be a module subprogram USED by, or declared as EXTERNAL in, the (sub)program from which D02NBF is called. Parameters denoted as *Input* must **not** be changed by this procedure.

16: WKJAC(NWKJAC) – REAL (KIND=nag_wp) array *Communication Array*

17: NWKJAC – INTEGER *Input*

On entry: the dimension of the array WKJAC as declared in the (sub)program from which D02NBF is called. This value must be the same as that supplied to the linear algebra setup routine D02NSF.

Constraint: $\text{NWKJAC} \geq \text{LDYSAV} \times (\text{LDYSAV} + 1)$.

18: MONITR – SUBROUTINE, supplied by the NAG Library or the user. *External Procedure*

MONITR performs tasks requested by you. If this option is not required, then the actual argument for MONITR must be the dummy routine D02NBY. (D02NBY is included in the NAG Library.)

The specification of MONITR is:

```
SUBROUTINE MONITR (NEQ, LDYSAV, T, HLAST, HNEXT, Y, YDOT, YSAV, R,      &
                  ACOR, IMON, INLN, HMIN, HMAX, NQU)
```

```
INTEGER          NEQ, LDYSAV, IMON, INLN, NQU
```

```
REAL (KIND=nag_wp) T, HLAST, HNEXT, Y(NEQ), YDOT(NEQ),      &
                  YSAV(LDYSAV, sdysav), R(NEQ), ACOR(NEQ, 2), HMIN,      &
                  HMAX
```

where *sdysav* is the numerical value of SDYSAV in the call of D02NBF.

1:	NEQ – INTEGER <i>On entry:</i> the number of differential equations being solved.	<i>Input</i>
2:	LDYSAV – INTEGER <i>On entry:</i> an upper bound on the number of differential equations to be solved.	<i>Input</i>
3:	T – REAL (KIND=nag_wp) <i>On entry:</i> the current value of the independent variable.	<i>Input</i>
4:	HLAST – REAL (KIND=nag_wp) <i>On entry:</i> the last step size successfully used by the integrator.	<i>Input</i>
5:	HNEXT – REAL (KIND=nag_wp) <i>On entry:</i> the step size that the integrator proposes to take on the next step. <i>On exit:</i> the next step size to be used. If this is different from the input value, then IMON must be set to 4.	<i>Input/Output</i>
6:	Y(NEQ) – REAL (KIND=nag_wp) array <i>On entry:</i> y , the values of the dependent variables evaluated at t . <i>On exit:</i> these values must not be changed unless IMON is set to 2.	<i>Input/Output</i>
7:	YDOT(NEQ) – REAL (KIND=nag_wp) array <i>On entry:</i> the time derivatives y' of the vector y .	<i>Input</i>
8:	YSAV(LDYSAV, <i>sdysav</i>) – REAL (KIND=nag_wp) array <i>On entry:</i> workspace to enable you to carry out interpolation using either of the routines D02XJF or D02XKF.	<i>Input</i>
9:	R(NEQ) – REAL (KIND=nag_wp) array <i>On entry:</i> if IMON = 0 and INLN = 3, the first NEQ elements contain the residual vector, $y' - g(t, y)$.	<i>Input</i>
10:	ACOR(NEQ,2) – REAL (KIND=nag_wp) array <i>On entry:</i> with IMON = 1, ACOR(i , 1) contains the weight used for the i th equation when the norm is evaluated, and ACOR(i , 2) contains the estimated local error for the i th equation. The scaled local error at the end of a timestep may be obtained by calling the real function D02ZAF as follows: <pre>IFAIL = 1 ERRLOC = D02ZAF(NEQ, ACOR(1,2), ACOR(1,1), IFAIL) ! CHECK IFAIL BEFORE PROCEEDING</pre>	<i>Input</i>
11:	IMON – INTEGER <i>On entry:</i> a flag indicating under what circumstances MONITR was called: IMON = -2 Entry from the integrator after IRES = 4 (set in FCN) caused an early termination (this facility could be used to locate discontinuities). IMON = -1 The current step failed repeatedly. IMON = 0 Entry after a call to the internal nonlinear equation solver (see INLN).	<i>Input/Output</i>

IMON = 1

The current step was successful.

On exit: may be reset to determine subsequent action in D02NBF.

IMON = -2

Integration is to be halted. A return will be made from the integrator to the calling (sub)program with IFAIL = 12.

IMON = -1

Allow the integrator to continue with its own internal strategy. The integrator will try up to three restarts unless IMON is set $\neq -1$ on exit.

IMON = 0

Return to the internal nonlinear equation solver, where the action taken is determined by the value of INLN (see INLN).

IMON = 1

Normal exit to the integrator to continue integration.

IMON = 2

Restart the integration at the current time point. The integrator will restart from order 1 when this option is used. The solution Y, provided by MONITR, will be used for the initial conditions.

IMON = 3

Try to continue with the same step size and order as was to be used before the call to MONITR. HMIN and HMAX may be altered if desired.

IMON = 4

Continue the integration but using a new value of HNEXT and possibly new values of HMIN and HMAX.

12: INLN – INTEGER *Output*

On exit: the action to be taken by the internal nonlinear equation solver when MONITR is exited with IMON = 0. By setting INLN = 3 and returning to the integrator, the residual vector is evaluated and placed in the array R, and then MONITR is called again. At present this is the only option available: INLN must not be set to any other value.

13: HMIN – REAL (KIND=nag_wp) *Input/Output*

On entry: the minimum step size to be taken on the next step.

On exit: the minimum step size to be used. If this is different from the input value, then IMON must be set to 3 or 4.

14: HMAX – REAL (KIND=nag_wp) *Input/Output*

On entry: the maximum step size to be taken on the next step.

On exit: the maximum step size to be used. If this is different from the input value, then IMON must be set to 3 or 4. If HMAX is set to zero, no limit is assumed.

15: NQU – INTEGER *Input*

On entry: the order of the integrator used on the last step. This is supplied to enable you to carry out interpolation using either of the routines D02XJF or D02XKF.

MONITR must either be a module subprogram USED by, or declared as EXTERNAL in, the (sub)program from which D02NBF is called. Parameters denoted as *Input* must **not** be changed by this procedure.

19: ITASK – INTEGER

Input

On entry: the task to be performed by the integrator.

ITASK = 1

Normal computation of output values of $y(t)$ at $t = \text{TOUT}$ (by overshooting and interpolating).

ITASK = 2

Take one step only and return.

ITASK = 3

Stop at the first internal integration point at or beyond $t = \text{TOUT}$ and return.

ITASK = 4

Normal computation of output values of $y(t)$ at $t = \text{TOUT}$ but without overshooting $t = \text{TCRIT}$ (e.g., see D02MVF). TCRIT must be specified as an option in one of the integrator setup routines before the first call to the integrator, or specified in the optional input routine before a continuation call. TCRIT may be equal to or beyond TOUT, but not before it, in the direction of integration.

ITASK = 5

Take one step only and return, without passing TCRIT (e.g., see D02MVF). TCRIT must be specified as under ITASK = 4.

Constraint: ITASK = 1, 2, 3, 4 or 5.

20: ITRACE – INTEGER

Input

On entry: the level of output that is printed by the integrator. ITRACE may take the value -1 , 0 , 1 , 2 or 3 .

ITRACE < -1

-1 is assumed and similarly if ITRACE > 3 , then 3 is assumed.

ITRACE = -1

No output is generated.

ITRACE = 0

Only warning messages are printed on the current error message unit (see X04AAF).

ITRACE > 0

Warning messages are printed as above, and on the current advisory message unit (see X04ABF) output is generated which details Jacobian entries, the nonlinear iteration and the time integration. The advisory messages are given in greater detail the larger the value of ITRACE.

21: IFAIL – INTEGER

Input/Output

On entry: IFAIL must be set to 0 , -1 or 1 . If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, because for this routine the values of the output parameters may be useful even if IFAIL $\neq 0$ on exit, the recommended value is -1 . **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

On exit: IFAIL = 0 unless the routine detects an error or a warning has been flagged (see Section 6).

6 Error Indicators and Warnings

If on entry $IFAIL = 0$ or -1 , explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

$IFAIL = 1$

An illegal input was detected on entry, or after an internal call to MONITR. If $ITRACE > -1$, then the form of the error will be detailed on the current error message unit (see X04AAF).

$IFAIL = 2$

The maximum number of steps specified has been taken (see the description of optional inputs in the integrator setup routines and the optional input continuation routine, D02NZF).

$IFAIL = 3$

With the given values of RTOL and ATOL no further progress can be made across the integration range from the current point T. The components $Y(1), Y(2), \dots, Y(NEQ)$ contain the computed values of the solution at the current point T.

$IFAIL = 4$

There were repeated error test failures on an attempted step, before completing the requested task, but the integration was successful as far as T. The problem may have a singularity, or the local error requirements may be inappropriate.

$IFAIL = 5$

There were repeated convergence test failures on an attempted step, before completing the requested task, but the integration was successful as far as T. This may be caused by an inaccurate Jacobian matrix or one which is incorrectly computed.

$IFAIL = 6$

Some error weight w_i became zero during the integration (see the description of ITOL). Pure relative error control ($ATOL(i) = 0.0$) was requested on a variable (the i th) which has now vanished. The integration was successful as far as T.

$IFAIL = 7$

FCN set its error flag ($IRES = 3$) continually despite repeated attempts by the integrator to avoid this.

$IFAIL = 8$

Not used for this integrator.

$IFAIL = 9$

A singular Jacobian $\frac{\partial r}{\partial y}$ has been encountered. This error exit is unlikely to be taken when solving explicit ordinary differential equations. You should check the problem formulation and Jacobian calculation.

$IFAIL = 10$

An error occurred during Jacobian formulation or back-substitution (a more detailed error description may be directed to the current error message unit, see X04AAF).

IFAIL = 11

FCN signalled the integrator to halt the integration and return (IRES = 2). Integration was successful as far as T.

IFAIL = 12

MONITR set IMON = -2 and so forced a return but the integration was successful as far as T.

IFAIL = 13

The requested task has been completed, but it is estimated that a small change in RTOL and ATOL is unlikely to produce any change in the computed solution. (Only applies when you are not operating in one step mode, that is when ITASK \neq 2 or 5.)

IFAIL = 14

The values of RTOL and ATOL are so small that D02NBF is unable to start the integration.

IFAIL = 15

The linear algebra setup routine D02NSF was not called prior to calling D02NBF.

7 Accuracy

The accuracy of the numerical solution may be controlled by a careful choice of the parameters RTOL and ATOL, and to a much lesser extent by the choice of norm. You are advised to use scalar error control unless the components of the solution are expected to be poorly scaled. For the type of decaying solution typical of many stiff problems, relative error control with a small absolute error threshold will be most appropriate (that is, you are advised to choose ITOL = 1 with ATOL(1) small but positive).

8 Further Comments

The cost of computing a solution depends critically on the size of the differential system and to a lesser extent on the degree of stiffness of the problem. For D02NBF the cost is proportional to NEQ^3 , though for problems which are only mildly nonlinear the cost may be dominated by factors proportional to NEQ^2 except for very large problems.

In general, you are advised to choose the Backward Differentiation Formula option (setup routine D02NVF) but if efficiency is of great importance and especially if it is suspected that $\frac{\partial g}{\partial y}$ has complex eigenvalues near the imaginary axis for some part of the integration, you should try the BLEND option (setup routine D02NWF).

9 Example

This example solves the well-known stiff Robertson problem

$$\begin{aligned} a' &= -0.04a + 1.0E4bc \\ b' &= 0.04a - 1.0E4bc - 3.0E7b^2 \\ c' &= 3.0E7b^2 \end{aligned}$$

over the range $[0, 10]$ with initial conditions $a = 1.0$ and $b = c = 0.0$ using scalar error control (ITOL = 1) and computation of the solution at TOUT = 10.0 with TCRIT (e.g., see D02MVF) set to 10.0 (ITASK = 4). D02NBY is used for MONITR, a BDF integrator (setup routine D02NVF) is used and a modified Newton method is selected. This example illustrates the use of both a numerical and an analytical Jacobian.

9.1 Program Text

```

!   D02NBF Example Program Text
!   Mark 24 Release. NAG Copyright 2012.

Module d02nbfe_mod

!   D02NBF Example Program Module:
!       Parameters and User-defined Routines

!   .. Use Statements ..
Use nag_library, Only: nag_wp
!   .. Implicit None Statement ..
Implicit None
!   .. Parameters ..
Real (Kind=nag_wp), Parameter      :: alpha = 0.04_nag_wp
Real (Kind=nag_wp), Parameter      :: beta  = 1.0E4_nag_wp
Real (Kind=nag_wp), Parameter      :: gamma = 3.0E7_nag_wp
Real (Kind=nag_wp), Parameter      :: one   = 1.0_nag_wp
Real (Kind=nag_wp), Parameter      :: two   = 2.0_nag_wp
Integer, Parameter                  :: iset  = 1, itrace = 0, neq = 3,   &
                                     nin    = 5, nout  = 6
Integer, Parameter                  :: nrw   = 50 + 4*neq
Integer, Parameter                  :: nwkjac = neq*(neq+1)
Integer, Parameter                  :: ldysav = neq
Contains
Subroutine fcn(neq,t,y,f,ires)

!   .. Scalar Arguments ..
Real (Kind=nag_wp), Intent (In)      :: t
Integer, Intent (Inout)              :: ires
Integer, Intent (In)                 :: neq
!   .. Array Arguments ..
Real (Kind=nag_wp), Intent (Out)     :: f(neq)
Real (Kind=nag_wp), Intent (In)     :: y(neq)
!   .. Executable Statements ..
f(1) = -alpha*y(1) + beta*y(2)*y(3)
f(2) = alpha*y(1) - beta*y(2)*y(3) - gamma*y(2)*y(2)
f(3) = gamma*y(2)*y(2)
Return
End Subroutine fcn

Subroutine jac(neq,t,y,h,d,p)

!   .. Scalar Arguments ..
Real (Kind=nag_wp), Intent (In)     :: d, h, t
Integer, Intent (In)                :: neq
!   .. Array Arguments ..
Real (Kind=nag_wp), Intent (Inout)  :: p(neq,neq)
Real (Kind=nag_wp), Intent (In)     :: y(neq)
!   .. Local Scalars ..
Real (Kind=nag_wp)                  :: hxd
!   .. Executable Statements ..
hxd = h*d
p(1,1) = one - hxd*(-alpha)
p(1,2) = -hxd*(beta*y(3))
p(1,3) = -hxd*(beta*y(2))
p(2,1) = -hxd*(alpha)
p(2,2) = one - hxd*(-beta*y(3)-two*gamma*y(2))
p(2,3) = -hxd*(-beta*y(2))
!   Do not need to set P(3,1) since Jacobian preset to zero
!   P(3,1) = -HxD*(0.0E0)
p(3,2) = -hxd*(two*gamma*y(2))
p(3,3) = one - hxd*(0.0_nag_wp)
Return
End Subroutine jac
End Module d02nbfe_mod

Program d02nbfe

!   D02NBF Example Main Program

```

```

! .. Use Statements ..
Use nag_library, Only: d02nbf, d02nby, d02nbz, d02nsf, d02nvf, d02nyf, &
                        nag_wp, x04abf
Use d02nbfe_mod, Only: fcn, iset, itrace, jac, ldysav, neq, nin, nout, &
                        nrw, nwkjac
! .. Implicit None Statement ..
Implicit None
! .. Local Scalars ..
Real (Kind=nag_wp)          :: h, h0, hmax, hmin, hu, t, tcrit, &
                             tcur, tinit, tolsf, tout
Integer                    :: i, icalse, ifail, imxer, itask, &
                             itol, maxord, maxstp, mxhnil, &
                             niter, nje, nq, nqu, nre, nst, &
                             outchn, sdysav
Logical                    :: petzld
! .. Local Arrays ..
Real (Kind=nag_wp), Allocatable :: atol(:), rtol(:), rwork(:), &
                             wkjac(:), y(:), ydot(:), &
                             yinit(:), ysav(:, :)
Real (Kind=nag_wp)          :: con(6)
Integer                    :: inform(23)
Logical, Allocatable        :: algequ(:)
! .. Executable Statements ..
Write (nout,*) 'D02NBF Example Program Results'
! Skip heading in data file
Read (nin,*)
Read (nin,*) maxord, maxstp, mxhnil
sdysav = maxord + 1
Allocate (atol(neq),rtol(neq),rwork(nrw),wkjac(nwkjac),y(neq), &
         yinit(neq),ydot(neq),ysav(ldysav,sdysav),algequ(neq))
outchn = nout
Call x04abf(iset,outchn)

Read (nin,*) petzld
Read (nin,*) hmin, hmax, h0
Read (nin,*) tinit, tout
Read (nin,*) itol
Read (nin,*) yinit(1:neq)
Read (nin,*) rtol(1), atol(1)

! Two cases. In both cases:
!         integrate to tout without passing tout;
!         use B.D.F formulae with a Newton method;
!         use default values for the array con;
!         use scalar tolerances;
!         use NAG dummy routine D02NBY in place of MONITR subroutine.
con(1:6) = 0.0_nag_wp
tcrit = tout
itask = 4

cases: Do icalse = 1, 2

    t = tinit
    y(1:neq) = yinit(1:neq)

!         ifail: behaviour on error exit
!         =0 for hard exit, =1 for quiet-soft, =-1 for noisy-soft
ifail = 0
Call d02nvf(neq,sdysav,maxord,'Newton',petzld,con,tcrit,hmin,hmax,h0, &
           maxstp,mxhnil,'Average-L2',rwork,ifail)

Write (nout,*)
ifail = 0
Select Case (icalse)
Case (1)
!         First case. The Jacobian is evaluated internally.
Call d02nsf(neq,neq,'Numerical',nwkjac,rwork,ifail)

```

```

        Write (nout,*) ' Numerical Jacobian'
    Case (2)
!       Second case. The Jacobian is evaluated by jac.
        Call d02nsf(neq,neq,'Analytical',nwkjac,rwork,ifail)

        Write (nout,*) ' Analytic Jacobian'
    End Select

    Write (nout,99993)(i,i=1,neq)
    Write (nout,99999) t, y(1:neq)

!       Soft fail and error messages only

    ifail = -1
    Select Case (icase)
    Case (1)
        Call d02nbf(neq,ldysav,t,tout,y,ydot,rwork,rtol,atol,itol,inform, &
            fcn,ysav,sdysav,d02nbz,wkjac,nwkjac,d02nby,itask,itrace,ifail)
    Case (2)
        Call d02nbf(neq,ldysav,t,tout,y,ydot,rwork,rtol,atol,itol,inform, &
            fcn,ysav,sdysav,jac,wkjac,nwkjac,d02nby,itask,itrace,ifail)
    End Select

    If (ifail==0) Then
        Write (nout,99999) t, y(1:neq)

        ifail = 0
        Call d02nyf(neq,neq,hu,h,tcurl,tolsf,rwork,nst,nre,nje,nqu,nq,niter, &
            imxer,algequ,inform,ifail)

        Write (nout,*)
        Write (nout,99997) hu, h, tcurl
        Write (nout,99996) nst, nre, nje
        Write (nout,99995) nqu, nq, niter
        Write (nout,99994) ' Max Err Comp = ', imxer
        Write (nout,*)
    Else
        Write (nout,*)
        Write (nout,99998) 'Exit D02NBF with IFAIL = ', ifail, ' and T = ', &
            t
    End If
End Do cases

99999 Format (1X,F8.3,3(F13.5,2X))
99998 Format (1X,A,I2,A,E12.5)
99997 Format (1X,' HUSED = ',E12.5,' HNEXT = ',E12.5,' TCUR = ',E12.5)
99996 Format (1X,' NST = ',I6,' NRE = ',I6,' NJE = ',I6)
99995 Format (1X,' NQU = ',I6,' NQ = ',I6,' NITER = ',I6)
99994 Format (1X,A,I4)
99993 Format (/1X,' X ',3(' Y(',I1,') ')')
        End Program d02nbfe

```

9.2 Program Data

D02NBF Example Program Data

5	200	5	:	maxord, maxstp, mxhnil
.	FALSE.		:	petzld
1.0E-10	10.0	0.0	:	hmin, hmax, h0
0.0	10.0		:	t, tout
1			:	itol
1.0	0.0	0.0	:	y
1.0E-4	1.0E-7		:	rtol, atol

9.3 Program Results

D02NBF Example Program Results

Numerical Jacobian

X	Y(1)	Y(2)	Y(3)
0.000	1.00000	0.00000	0.00000
10.000	0.84136	0.00002	0.15863

HUSED = 0.51867E+00 HNEXT = 0.51867E+00 TCUR = 0.10000E+02
 NST = 55 NRE = 132 NJE = 17
 NQU = 3 NQ = 3 NITER = 79
 Max Err Comp = 3

Analytic Jacobian

X	Y(1)	Y(2)	Y(3)
0.000	1.00000	0.00000	0.00000
10.000	0.84136	0.00002	0.15863

HUSED = 0.51867E+00 HNEXT = 0.51867E+00 TCUR = 0.10000E+02
 NST = 55 NRE = 81 NJE = 17
 NQU = 3 NQ = 3 NITER = 79
 Max Err Comp = 3

